In[1]:=
```
ker[A_] := Module[{X = NullSpace[A]}, Print["ker = lin", MatrixForm[Transpose[X]]]; X]
```

In[2]:=
```
A = {{0, 0, 0, 0, 0, 6, 0}, {4, 4, 0, 3, 0, -4, 0}, {0, 0, 4, 0, 0, 0, 2}, {0, 0, 0, 4, 0, 0, 0},
   {4, 2, 0, 2, 4, -4, 0}, {-4, 0, 0, -1, 0, 10, 0}, {4, 0, 0, 5, 0, -4, 4}};
MatrixForm[
 A]
```

Out[3]//MatrixForm=

$$\begin{pmatrix} 0 & 0 & 0 & 0 & 0 & 6 & 0 \\ 4 & 4 & 0 & 3 & 0 & -4 & 0 \\ 0 & 0 & 4 & 0 & 0 & 0 & 2 \\ 0 & 0 & 0 & 4 & 0 & 0 & 0 \\ 4 & 2 & 0 & 2 & 4 & -4 & 0 \\ -4 & 0 & 0 & -1 & 0 & 10 & 0 \\ 4 & 0 & 0 & 5 & 0 & -4 & 4 \end{pmatrix}$$

In[4]:=
```
(* wielomian charakterystyczny *)
chi[t] = Det[A - t IdentityMatrix[7]]
```

Out[4]= $24576 - 40960\,t + 29184\,t^2 - 11520\,t^3 + 2720\,t^4 - 384\,t^5 + 30\,t^6 - t^7$

In[5]:=
```
Factor[chi[t]]
```

Out[5]= $-(-6+t)\,(-4+t)^6$

In[6]:=
```
(* wartosc wlasna 6 *)
B6 = A - 6 IdentityMatrix[7]; MatrixForm[B6]
K = ker[B6];
```

Out[6]//MatrixForm=

$$\begin{pmatrix} -6 & 0 & 0 & 0 & 0 & 6 & 0 \\ 4 & -2 & 0 & 3 & 0 & -4 & 0 \\ 0 & 0 & -2 & 0 & 0 & 0 & 2 \\ 0 & 0 & 0 & -2 & 0 & 0 & 0 \\ 4 & 2 & 0 & 2 & -2 & -4 & 0 \\ -4 & 0 & 0 & -1 & 0 & 4 & 0 \\ 4 & 0 & 0 & 5 & 0 & -4 & -2 \end{pmatrix}$$

$$\text{ker} = \text{lin}\begin{pmatrix} 1 \\ 0 \\ 0 \\ 0 \\ 0 \\ 1 \\ 0 \end{pmatrix}$$

In[8]:=
```
u1 = K[[1]]
```

Out[8]= {1, 0, 0, 0, 0, 1, 0}

In[9]:=
```
(* wartosc wlasna 4 *)
B4 = A - 4 IdentityMatrix[7]; MatrixForm[B4]
K1 = ker[B4];
```

Out[9]//MatrixForm=

$$\begin{pmatrix} -4 & 0 & 0 & 0 & 0 & 6 & 0 \\ 4 & 0 & 0 & 3 & 0 & -4 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 2 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 4 & 2 & 0 & 2 & 0 & -4 & 0 \\ -4 & 0 & 0 & -1 & 0 & 6 & 0 \\ 4 & 0 & 0 & 5 & 0 & -4 & 0 \end{pmatrix}$$

$$\text{ker} = \text{lin}\begin{pmatrix} 0 & 0 \\ 0 & 0 \\ 0 & 1 \\ 0 & 0 \\ 1 & 0 \\ 0 & 0 \\ 0 & 0 \end{pmatrix}$$

In[11]:=
```
MatrixForm[B4.B4]
K2 = ker[B4.B4];
```

Out[11]//MatrixForm=

$$\begin{pmatrix} -8 & 0 & 0 & -6 & 0 & 12 & 0 \\ 0 & 0 & 0 & 4 & 0 & 0 & 0 \\ 8 & 0 & 0 & 10 & 0 & -8 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 8 & 0 & 0 & 10 & 0 & -8 & 0 \\ -8 & 0 & 0 & -6 & 0 & 12 & 0 \\ 0 & 0 & 0 & 4 & 0 & 0 & 0 \end{pmatrix}$$

$$\text{ker} = \text{lin}\begin{pmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 1 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 1 & 0 & 0 & 0 \end{pmatrix}$$

In[13]:=
```
MatrixForm[B4.B4.B4]
K3 = ker[B4.B4.B4];
```

Out[13]//MatrixForm=

$$\begin{pmatrix} -16 & 0 & 0 & -12 & 0 & 24 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 8 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 8 & 0 & 0 & 0 \\ -16 & 0 & 0 & -12 & 0 & 24 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 \end{pmatrix}$$

$$\text{ker} = \text{lin}\begin{pmatrix} 0 & 3 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 1 \\ 0 & 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 & 0 \\ 0 & 2 & 0 & 0 & 0 \\ 1 & 0 & 0 & 0 & 0 \end{pmatrix}$$

In[15]:=
```
MatrixForm[B4.B4.B4.B4]
K4 = ker[B4.B4.B4.B4];
```

Out[15]//MatrixForm=

$$\begin{pmatrix} -32 & 0 & 0 & -24 & 0 & 48 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ -32 & 0 & 0 & -24 & 0 & 48 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 \end{pmatrix}$$

$$\text{ker} = \text{lin}\begin{pmatrix} 0 & 3 & 0 & -3 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 1 \\ 0 & 0 & 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 4 & 0 & 0 \\ 0 & 0 & 1 & 0 & 0 & 0 \\ 0 & 2 & 0 & 0 & 0 & 0 \\ 1 & 0 & 0 & 0 & 0 & 0 \end{pmatrix}$$

In[22]:=
```
(* najdluzszy lancuszek *)
v4 = K4[[4]]
v3 = B4.v4
v2 = B4.v3
v1 = B4.v2
B4.v1
```

Out[22]= {-3, 0, 0, 4, 0, 0, 0}

Out[23]= {12, 0, 0, 0, -4, 8, 8}

Out[24]= {0, 16, 16, 0, 16, 0, 16}

Out[25]= {0, 0, 32, 0, 32, 0, 0}

Out[26]= {0, 0, 0, 0, 0, 0, 0}

In[27]:=
```
(* Z wektorow K1=ker(B4) trzeba wybrac koncowy element drugiego lanscuszka,
aby wraz z v1 rozpinal K1. Ta metoda wymaga pozniej szukania przeciwobrazu v1*)
K1
```

Out[27]= {{0, 0, 0, 0, 1, 0, 0}, {0, 0, 1, 0, 0, 0, 0}}

In[28]:=
```
w1 = K1[[1]]
```

Out[28]= {0, 0, 0, 0, 1, 0, 0}

In[29]:=
```
(* zastanawiamy sie czego obrazem jest w1; przypomnijmy jaka jest macierz B4. *)
MatrixForm[B4]
```

Out[29]//MatrixForm=

$$\begin{pmatrix} -4 & 0 & 0 & 0 & 0 & 6 & 0 \\ 4 & 0 & 0 & 3 & 0 & -4 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 2 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 4 & 2 & 0 & 2 & 0 & -4 & 0 \\ -4 & 0 & 0 & -1 & 0 & 6 & 0 \\ 4 & 0 & 0 & 5 & 0 & -4 & 0 \end{pmatrix}$$

In[32]:=
```
(* zauwazamy, ze w1=epsilon5 jest obrazem *)
w2 = {0, 1 / 2, 0, 0, 0, 0, 0};
w1 == B4.w2
```

Out[33]= True

In[34]:=
```
(* macierz przejscia z Jordana do standardowej *)
X = Transpose[{u1, v1, v2, v3, v4, w1, w2}];
MatrixForm[X]
```

Out[35]//MatrixForm=

$$\begin{pmatrix} 1 & 0 & 0 & 12 & -3 & 0 & 0 \\ 0 & 0 & 16 & 0 & 0 & 0 & \frac{1}{2} \\ 0 & 32 & 16 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 4 & 0 & 0 \\ 0 & 32 & 16 & -4 & 0 & 1 & 0 \\ 1 & 0 & 0 & 8 & 0 & 0 & 0 \\ 0 & 0 & 16 & 8 & 0 & 0 & 0 \end{pmatrix}$$

In[36]:=
```
(* postac Jordana *)
MatrixForm[Inverse[X].A.X]
```

Out[36]//MatrixForm=

$$\begin{pmatrix} 6 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 4 & 1 & 0 & 0 & 0 & 0 \\ 0 & 0 & 4 & 1 & 0 & 0 & 0 \\ 0 & 0 & 0 & 4 & 1 & 0 & 0 \\ 0 & 0 & 0 & 0 & 4 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 4 & 1 \\ 0 & 0 & 0 & 0 & 0 & 0 & 4 \end{pmatrix}$$